1 Hypothesis Testing

Running Example: Bone Marrow Transplant

- At the time of transplant, each patient is classified into one of three risk categories: ALL (acute lymphoblastic leukemia), AML (acute myelocytic leukemia)-Low Risk, and AML- High Risk. The endpoint of interest is the disease-free survival time, which is the time to death or relapse or to the end of the study in days.

```
* output graph;
ods listing gpath='/folders/myfolders/Lab6/';
ods graphics / imagename="p1" imagefmt=png;
proc lifetest data=bone_marrow plots=survival(atrisk=0 to 2500 by 500);
time t2*dfree(0);
strata g;
run;
```



- Patients in the AML-Low Risk group experience a longer disease-free survival than those in the ALL group, who in turn fare better than those in the AML-High Risk group.

Q: Test the hypothesis that the disease-free survival functions of these three populations are the same over the range of observation, $t \leq 2204$ days, versus the alternative that at least one of the populations has a different survival rate.

K-Sample Test $H_0 : h_1(t) = h_2(t) = \cdots = h_K(t)$, for all $t \leq \tau$, versus H_A : at least one of the $h_j(t)$'s is different for some $t \leq \tau$.

- $\chi^2 = (Z_1(\tau), \ldots, Z_{K-1}(\tau)) \, \Sigma^{-1} \, (Z_1(\tau), \ldots, Z_{K-1}(\tau))^t$
- When the null hypothesis is true, this statistic has a chi-squared distribution, for large samples with $K - 1$ degrees of freedom.
- ```
  * K-sample test;
  proc lifetest data=bone_marrow plots=survival(atrisk=0 to 2500 by 500);
  time t2*dfree(0);
  strata g/test=(logrank TARONE PETO MODPETO
  FLEMING(0,1)    );
  run;
  ```

Test of Equality over Strata			
Test	Chi-Square	DF	Pr > Chi-Square
Log-Rank	13.8037	2	0.0010
Tarone	15.6529	2	0.0004
Peto	15.7260	2	0.0004
Modified Peto	15.7781	2	0.0004
Fleming(0,1)	6.1097	2	0.0471

- All of these tests agree with the conclusion that the disease-free survival curves are not the same in these three disease categories.
  ```

Tests for Trend to detect ordered alternatives:

$$H_0 : h_1(t) = h_2(t) = \cdots = h_K(t), \text{ for } t \leq \tau$$

against $H_A : h_1(t) \leq h_2(t) \leq \cdots \leq b_K(t)$ for $t \leq \tau$, with at least on strict inequality. The alternative hypothesis is equivalent to the hypothesis that $S_1(t) \geq S_2(t) \geq \cdots \geq S_K(t)$

- To construct the test, a sequence of scores $a_1 < a_2 < \cdots < a_K$ is selected. Any increasing set of scores can be used in constructing the test. In most cases, the scores $a_j = j$ are used.
- The test statistic is $Z = \frac{\sum_{j=1}^{K} a_j Z_j(\tau)}{\sqrt{\sum_{j=1}^{K} \sum_{g=1}^{K} a_j a_g \hat{\sigma}_{jg}}}$
- ```
  * Tests for trend;
  proc lifetest data=bone_marrow plots=survival(atrisk=0 to 2500 by 500);
  time t2*dfree(0);
  strata g/trend test=(logrank TARONE PETO MODPETO
  FLEMING(0,1)    );
  run;
  ```
- "If there is only one STRATA variable and the variable is numeric, the unformatted values of the variable are used as the scores; otherwise, the scores are 1, 2, ... , in the given order of the strata."

Scores for Trend Test

g	Score
ALL	1
AML High Risk	2
AML Low Risk	3

Trend Tests

Test	TestStatistic	Standard Error	z-Score	Pr > |z|	Pr < z	Pr > z
Log-Rank	-17.1144	7.5489	-2.2672	0.0234	0.0117	0.9883
Tarone	-170.5481	73.7041	-2.3140	0.0207	0.0103	0.9897
Peto	-12.0750	5.4065	-2.2334	0.0255	0.0128	0.9872
Modified Peto	-11.9632	5.3532	-2.2348	0.0254	0.0127	0.9873
Fleming(0,1)	-4.9556	2.6027	-1.9040	0.0569	0.0285	0.9715
  ```

Stratified Tests Risk factors/Confounders

- we found that there was evidence of a difference in disease-free survival rates between bone marrow patients with ALL, low-risk AML and high-risk AML. A proper comparison of these three disease groups should account for other factors which may influence disease-free survival. One such factor is the use of a graft-versus-host prophylactic combining methotretexate (MTX) with some other agent.
- We assume that our test is to be stratified on M levels of a set of covariates.
- $H_0 : b_{1s}(t) = b_{2s}(t) = \cdots = b_{Ks}(t)$, for $s = 1, \ldots, M, t < \tau$
- $Z_{j.}(\tau) = \sum_{s=1}^{M} Z_{js}(\tau)$
- Test statistics: $(Z_{1.}(\tau), \ldots, Z_{K-1.}(\tau))\, \Sigma_{\cdot}^{-1}\, (Z_{1.}(\tau), \ldots, Z_{K-1.}(\tau))^t$ Chi-square with df=K-1
- ```
  * Stratefied test;
  proc lifetest data=bone_marrow ;
  time t2*dfree(0);
  strata g/group=z10 test=(logrank TARONE PETO MODPETO
  FLEMING(0,1)    );
  run;
  ```

Stratified Test of Equality over Group			
Test	Chi-Square	DF	Pr > Chi-Square
Log-Rank	2.0600	1	0.1512
Tarone	3.2590	1	0.0710
Peto	4.5356	1	0.0332
Modified Peto	4.6107	1	0.0318
Fleming(0,1)	0.0907	1	0.7632
  ```

2 Semiparametric Proportional Hazards Regression

Interested in comparing two or more groups of times-to event, when covariates (explanatory variables, confounders, risk factors, independent variables) are present, then adjusting for the, the comparison between survival times between groups would be less biased and more efficient than simple nonparametric comparison.

- Data : $(T_j, \delta_j, \mathbf{Z}_j(t)), j = 1, \ldots, n$
- fixed-covariate case where $\mathbf{Z}_j(t) = \mathbf{Z}_j = (Z_{j1}, \ldots, Z_{jp})^t$
- $b(t \mid \mathbf{Z}) = b_0(t) \exp\left(\boldsymbol{\beta}^t \mathbf{Z}\right) = b_0(t) \exp\left(\sum_{k=1}^{p} \beta_k Z_k\right)$
- hazard rates are proportional: $\frac{b(t|\mathbf{Z})}{b(t|\mathbf{Z}^*)} = \frac{b_0(t) \exp\left[\sum_{k=1}^{p} \beta_k Z_k\right]}{b_0(t) \exp\left[\sum_{k=1}^{p} \beta_k Z_k^*\right]} = \exp\left[\sum_{k=1}^{p} \beta_k \left(Z_k - Z_k^*\right)\right]$
- relative risk (hazard ratio) of an individual with risk factor $\mathbf{Z}$ having the event as compared to an individual with risk factor $\mathbf{Z}^*$.
- if Z_1 indicates the treatment effect ($Z_1 = 1$ if treatment and $Z_1 = 0$ if placebo) and all other covariates have the same value, then, $b(t \mid \mathbf{Z})/\hat{h}\left(t \mid \mathbf{Z}^*\right) = \exp\left(\beta_1\right)$, is the risk of having the event if the individual received the treatment relative to the risk of having the event should the individual have received the placebo.

Example 1 Bone Marrow Transplant

- we wish to perform a proportional hazards regression with group as the single covariate in the model.
- ```
  proc phreg data =bone_marrow plots(overlay)=(survival);
  class g(desc);
  model t2*dfree(0) = g;
  run;
  ```

Analysis of Maximum Likelihood Estimates

Parameter		DF	Parameter Estimate	Standard Error	Chi-Square	Pr > ChiSq	Hazard Ratio	Label
g	AML Low Risk	1	-0.57418	0.28730	3.9942	0.0457	0.563	Disease Group AML Low Risk
g	AML High Risk	1	0.38262	0.26738	2.0478	0.1524	1.466	Disease Group AML High Risk

- the relative risk of relapse or dying for an AML-high risk patient relative to an an ALL patient is exp(0.38262)=1.466. That is, a patient who is high risk is 1.466 times more likely to die than an ALL patient.
- A test of the global hypothesis: $H_0 : \boldsymbol{\beta} = \boldsymbol{\beta}_0$
  ```

Testing Global Null Hypothesis: BETA=0			
Test	Chi-Square	DF	Pr > ChiSq
Likelihood Ratio	13.4307	2	0.0012
Score	13.7821	2	0.0010
Wald	13.0092	2	0.0015

- Adjusting for patient age and donor age
- ```
  proc phreg data =bone_marrow plots(overlay)=(survival);
  class g(desc);
  model t2*dfree(0) = g z1 z2;
  run;
  ```

Analysis of Maximum Likelihood Estimates								
Parameter		DF	Parameter Estimate	Standard Error	Chi-Square	Pr > ChiSq	Hazard Ratio	Label
g	AML Low Risk	1	-0.61543	0.30427	4.0910	0.0431	0.540	Disease Group AML Low Risk
g	AML High Risk	1	0.31252	0.29255	1.1412	0.2854	1.367	Disease Group AML High Risk
z1		1	0.07075	0.17795	0.1581	0.6909	1.073	Patient Age
z2		1	0.00244	0.01762	0.0191	0.8900	1.002	Donor Age

- relative risk for a 50-year-old patient compared to a 40-yearold (both with AML high risk or low risk or ALL) is exp(0.07) =1.073, but is not significant (p-value=0.69). Another way of stating the interpretation of a partial relative risk is that a 50-year-old patient has a probability of dying 1.073 times greater than the probability of dying for a 40-year-old patient with the same severity of disease. Same for donor age.
- Acquiring more than one curve, whether survival or hazard, after Cox regression in SAS requires use of the baseline statement in conjunction with the creation of a small dataset of covariate values at which to estimate our curves of interest.
- ```
  * survival plot;
  data covs;
  format g gs.;
  input g z1 z2 ;
  datalines;
  3 20 30
  3 30 30
  3 40 30
  3 50 30
  3 60 30
  ;
  run;
  ```

```
ods listing gpath='/folders/myfolders/Lab6/';
ods graphics / imagename="p1" imagefmt=png;
proc phreg data =bone_marrow plots(overlay)=(survival);
class g(desc);
model t2*dfree(0) = g z1 z2;
baseline covariates=covs out=base /rowid=z1;
run;
```



3 HW2 Q3

```
data q1;
input t c;
datalines;
3 0
4 0
5 1
6 0
6 1
8 1
11 0
14 0
15 0
16 1
;
run;
proc lifetest data=q1 nelson  ;
time t*c(1);
ods output ProductLimitEstimates=ProductLimitEstimates;
run;

data ProductLimitEstimates;
set ProductLimitEstimates;
if censor ne 1;

* variance for KM for survival;
VKM=stderr**2;

* H(t) via KM;
HKM=-log(Survival);
VHKM=(1/survival**2)*VKM;
keep t survival VKM HKM VHKM;
proc print;
run;
```

Obs	t	Survival	VKM	HKM	VHKM
1	0.0000	1.0000	0.000000	0.00000	0.00000
2	3.0000	0.9000	0.009000	0.10536	0.01111
3	4.0000	0.8000	0.016000	0.22314	0.02500
4	6.0000	0.6857	0.022950	0.37729	0.04881
5	11.0000	0.5143	0.034950	0.66498	0.13214
6	14.0000	0.3429	0.035125	1.07044	0.29881
7	15.0000	0.1714	0.023475	1.76359	0.79881